

9.2.1 Beispiele zum Schmidtschen Orthogonalisierungsverfahren

```
<<LinearAlgebra`Orthogonalization`
```

Legendrepolynome $P_n(x)$

■ Orthogonalisierung der Potenzen

Orthogonalisiert man die Potenzen von x im Intervall $[-1,1]$, ergeben sich die Legendrepolynome $P_n(x)$.

```
lp = Table[x^k, {k, 0, 7}]
{1, x, x^2, x^3, x^4, x^5, x^6, x^7}

lgs = GramSchmidt[lp, InnerProduct -> (Integrate[#1 #2, {x, -1, 1}] &)] // Simplify

{1/Sqrt[2], Sqrt[3/2] x, 1/2 Sqrt[5/2] (-1 + 3 x^2), 1/2 Sqrt[7/2] x (-3 + 5 x^2),
 3 (3 - 30 x^2 + 35 x^4)/(8 Sqrt[2]), 1/8 Sqrt[11/2] x (15 - 70 x^2 + 63 x^4),
 1/16 Sqrt[13/2] (-5 + 105 x^2 - 315 x^4 + 231 x^6), 1/16 Sqrt[15/2] x (-35 + 315 x^2 - 693 x^4 + 429 x^6)}

Table[LegendreP[k, x], {k, 0, Length[lp]-1}]//Simplify

{1, x, 1/2 (-1 + 3 x^2), 1/2 x (-3 + 5 x^2), 1/8 (3 - 30 x^2 + 35 x^4), 1/8 x (15 - 70 x^2 + 63 x^4),
 1/16 (-5 + 105 x^2 - 315 x^4 + 231 x^6), 1/16 x (-35 + 315 x^2 - 693 x^4 + 429 x^6)}

lpp = Table[Normalize[LegendreP[k, x],
  InnerProduct ->
    (Integrate[#1 #2, {x, -1, 1}] &)], {k, 0, Length[lp]-1}]//Simplify;
lgs == lpp

True
```

■ Berechnung der Orthogonalität und Normierung mittels der erzeugenden Funktion

Die erzeugende Funktion der Legendre - Polynome ist :

$$\frac{1}{\sqrt{1 - 2 x \alpha + \alpha^2}} = \sum_{n=0}^{\infty} \alpha^n P_n(x)$$

```

fa = 1 / Sqrt[1 - 2 α x + α^2]

1
-----
Sqrt[1 - 2 x α + α^2]

fb = fa /. α → β

1
-----
Sqrt[1 - 2 x β + β^2]

in = Integrate[fa fb, {x, -1, 1}]

If[ (Re[1/α + α] ≤ -2 || Re[1/α + α] ≥ 2 || Im[1/α + α] ≠ 0) &&
(Re[1/β + β] ≤ -2 || Re[1/β + β] ≥ 2 || Im[1/β + β] ≠ 0),
1/(2 Sqrt[α] Sqrt[β]) (Log[(2 Sqrt[(-1 + α)^2] Sqrt[α] Sqrt[(-1 + β)^2] Sqrt[β] - β - α^2 β - α (1 - 4 β + β^2))/Sqrt[α] Sqrt[β]] -
Log[(-β - α^2 β + 2 Sqrt[α] Sqrt[(1 + α)^2] Sqrt[β] Sqrt[(1 + β)^2] - α (1 + 4 β + β^2))/Sqrt[α] Sqrt[β]])],
Integrate[1/(Sqrt[1 - 2 x α + α^2] Sqrt[1 - 2 x β + β^2]), {x, -1, 1},
Assumptions → ! ((Re[1/α + α] ≤ -2 || Re[1/α + α] ≥ 2 || Im[1/α + α] ≠ 0) &&
(Re[1/β + β] ≤ -2 || Re[1/β + β] ≥ 2 || Im[1/β + β] ≠ 0))]

Simplify[in, 1 > α > 0 && 1 > β > 0]

1
-----
2 Sqrt[α β] (-Log[(-α - β - 4 α β - α^2 β - α β^2 + 2 (1 + α) Sqrt[α β] (1 + β))/Sqrt[α β]] +
Log[(-β - α^2 β + 2 Sqrt[(-1 + α)^2 α (-1 + β)^2 β] - α (1 - 4 β + β^2))/Sqrt[α β]])

```

Händische Integration des obigen Integrals gibt:

$$\begin{aligned}
& \int_{-1}^1 \frac{1}{\sqrt{1 - 2 x \alpha + \alpha^2}} \frac{1}{\sqrt{1 - 2 x \beta + \beta^2}} dx = \frac{1}{\sqrt{\alpha \beta}} \ln \frac{1 + \sqrt{\alpha \beta}}{1 - \sqrt{\alpha \beta}} = \\
& = \sum_{n=0}^{\infty} \frac{2}{2n+1} (\alpha \beta)^n = \sum_{n=0}^{\infty} \alpha^n \sum_{k=0}^{\infty} \beta^k \int_{-1}^1 dx P_n(x) P_k(x)
\end{aligned}$$